

NBSIR 86-3317

# Directional Hurricane Wind Speeds

A11102 510776

NATL INST OF STANDARDS & TECH R.I.C.



A11102510776

Hendrickson, Erik M/Directional hurrican  
QC100 .U56 NO.86-3317 V1986 C.2 NBS-PUB-

Erik M. Hendrickson

and

Emil Simiu

U.S. DEPARTMENT OF COMMERCE  
National Bureau of Standards  
National Engineering Laboratory  
Center for Building Technology  
Gaithersburg, MD 20899

September 1985

Issued February 1986

QC  
100  
U56  
86-3317  
1986  
C. 2

pared for:  
**clear Regulatory Commission**  
**shington, DC 20555**



CIRC-NBS

QC100

.U56

NO. 86-3317

1986

C.2

NBSIR 86-3317

**DIRECTIONAL HURRICANE WIND SPEEDS**

Erik M. Hendrickson

and

Emil Simiu

U.S. DEPARTMENT OF COMMERCE  
National Bureau of Standards  
National Engineering Laboratory  
Center for Building Technology  
Gaithersburg, MD 20899

September 1985

Issued February 1986

Prepared for:  
Nuclear Regulatory Commission  
Washington, DC 20555



---

**U.S. DEPARTMENT OF COMMERCE, Malcolm Baldrige, *Secretary***  
**NATIONAL BUREAU OF STANDARDS, Ernest Ambler, *Director***



## ABSTRACT

This report presents a simple procedure for estimating coastline hurricane wind speeds corresponding to any specified set of wind directions and to any specified mean recurrence interval. The procedure uses simulated directional hurricane wind speed data as described in the report Hurricane Wind Speeds in the United States (NBS BSS 124). These data are encoded on magnetic tape for 56 mileposts located along the U.S. Gulf and Atlantic coasts. The input to the procedure consists of the location along the coastline and the set of wind directions of interest. The output consists of the fastest 60-sec wind speeds at 10m above ground in open terrain corresponding to the set of directions specified in the input and to various mean recurrence intervals. In addition, the output includes the parameters of the Weibull distribution that best fits the hurricane wind speeds.

Key Words: Climatology; hurricanes; statistical analysis; structural engineering; wind engineering



## TABLE OF CONTENTS

1. INTRODUCTION
2. PROCEDURE FOR ESTIMATING HURRICANE WIND SPEEDS
3. DESCRIPTION OF COMPUTER PROGRAM
4. ACKNOWLEDGMENTS
5. REFERENCES
6. APPENDIX - PROGRAM LISTING AND EXAMPLES

## LIST OF TABLES

- Table 1. Compass Directions and Computer Code Numbers

## LIST OF FIGURES

- Figure 1. Locator Map
- Figure A1. Probability Plot, Directions 1-16
- Figure A2. Probability Plot, Directions 3-8
- Figure A3. Probability Plot, Direction 2
- Figure A4. Probability Plot, Directions 12-4





## 1. INTRODUCTION

It has previously been shown that predictions of extreme wind speeds in hurricane-prone regions cannot, in general, be based upon a statistical analysis of the largest wind speeds recorded at a given site [1, p. 89]. For this reason, estimates of extreme hurricane wind speeds at a site are commonly obtained by indirect methods [2,3,4] from (1) statistical information on the climatological characteristics of hurricanes, and (2) a physical model of the hurricane wind structure.

This indirect approach was described in some detail in Ref. 5, where it was applied to estimate hurricane wind speeds affecting the United States Gulf and Atlantic coasts. These estimates were subsequently used by the Subcommittee on Wind Loads, American National Standards Committee A58 on Building Code Requirements for Minimum Design Loads on Buildings, to develop the basic wind speed map included in Ref. 6. The estimated design wind speeds in Ref. 5 pertain to (1) hurricane winds blowing from any compass direction and (2) hurricane winds blowing from any direction within a  $22.5^\circ$  "half-octant" centered about one of the 16 main compass directions - N, NNE, NE, ENE, E, ESE, SE, SSE, S, SSW, SW, WSW, W, WNW, NW, and NNW.

The licensing review process for nuclear power plant facilities may require the estimation of hurricane winds blowing from any specified set of directions. This need arises where a certain wind effect is a function of wind direction (e.g., an aerodynamic load, the storm surge in an estuary, and so forth). The purpose of this report and of the attendant computer tape is to make available to the U.S. Nuclear Regulatory Commission a procedure that allows this estimation to be carried out rapidly and conveniently.

## 2. PROCEDURE FOR ESTIMATING HURRICANE WIND SPEEDS

We seek the distribution of annual extreme hurricane wind speeds,  $F_1(v)$ , so that any  $N$ -yr mean recurrence interval wind speed,  $v_N$ , can be estimated as  $v_N = F_1^{-1}(1 - 1/N)$ . However, the rate of occurrence of hurricanes at a particular site is almost always less than one per year. For this reason,  $F_1(v)$  is constructed by assuming the hurricane arrivals to be Poisson distributed, as described below.

The probability that  $V$  is less than  $v$  in  $\tau$  years,  $P_\tau(V < v)$ , is equal to the probability that  $V$  is less than  $v$  in  $n$  storms,  $P_n(V < v)$ , times the probability that  $n$  storms occur in  $\tau$  years,  $P(n, \tau)$ , summed over all  $n$ . Thus:

$$P_\tau(V < v) = \sum_{n=0}^{\infty} P_n(V < v) P(n, \tau) \quad (1)$$

The left hand side of Eq. 1 is the cumulative distribution of extreme hurricane wind speeds in  $\tau$  years:

$$P_\tau(V < v) = F_\tau(v) \quad (2)$$

Assuming that hurricane wind speeds are statistically independent from storm to storm,  $P_n(V < v)$  is:

$$P_n(V < v) = [P_s(V < v)]^n \quad (3)$$

in which  $P_s(V < v)$  is the probability that  $V$  is less than  $v$  for one storm. Note that  $P_s(V < v)$  is simply the cumulative distribution of wind speeds for one hurricane:

$$P_s(V < v) = F_s(v) \quad (4)$$

The probability that  $n$  storms occur in  $\tau$  years,  $P(n, \tau)$ , is assumed to have a Poisson distribution:

$$P(n, \tau) = (\mu\tau)^n \exp(-\mu\tau) / n! \quad (5)$$

in which  $\mu$  = rate of occurrence of hurricanes at the mile post, and in the directions, of interest. For example, if the rate of hurricane occurrence at a milepost is  $\mu_0$ , and if  $m$  hurricanes are simulated,  $k$  of which have nonzero wind speeds in the

directions of interest, then  $\mu = \mu_0 k/m$ . Combining Eqs. 2-5, Eq. 1 is rewritten as:

$$F_T(v) = \sum_{n=0}^{\infty} [F_S(v)]^n (\mu\tau)^n \exp(-\mu\tau) / n! \quad (6)$$

or, by noting that  $\sum_{n=0}^{\infty} x^n / n! = \exp(x)$ ,

$$F_T(v) = \exp[-\mu\tau(1 - F_S(v))] \quad (7)$$

At this point, two approaches are possible. First, we can assume that  $F_T(v)$  plots linearly on Weibull probability paper, but with an intercept shifted away from the origin. Indeed, for  $v = 0$ ,  $F_S(v) = 0$  so that  $F_T(v) = \exp(-\mu\tau)$ . In this case, the speeds  $v_i$  are plotted against  $\exp[-(\mu\tau)(m+1-i)/(m+1)]$  and the wind speed with mean recurrence interval  $N$  is:

$$v_N = F_T^{-1}(1 - 1/N) \quad (8)$$

A second approach is to assume that  $F_S(v)$  has a Weibull distribution. In this case,  $v_i$  is plotted against  $i/(m+1)$  and the wind speed with mean recurrence interval  $N$  is:

$$v_N = F_S^{-1}[1 + (1/\mu)\ln(1 - 1/N)] \quad (9)$$

A comparison of results obtained using Eqs. 8 and 9 for milepost 2400 and various directions indicated that there is no significant difference between the two approaches. Since the first approach was the one originally developed in connection with the work reported in Ref. 5, it was used in the computer program described in the following section.

### 3. DESCRIPTION OF COMPUTER PROGRAM

The purpose of the computer program presented herein is to estimate coastline hurricane wind speeds corresponding to any specified set of wind directions and to any specified mean recurrence interval.

The program is written in FORTRAN and consists of a main routine and two subroutines called WEIB and SORT. A listing of the program appears in the Appendix. The program has been written on magnetic tape along with 56 data files containing the simulated directional hurricane wind speed data. These data were obtained by using the procedure and information described in Ref. 5.

A locator map for the Gulf and Atlantic coasts, where mileposts in nautical miles are shown for the various locations along the coasts, is given in Fig. 1. Corresponding to each milepost is a data file containing hurricane wind speeds from 16 directions for 999 simulated hurricanes. The program user interested in estimating wind speeds at the location of the third milepost would attach the third data file to the program run.

The input to the computer program consists of the location along the coastline and the set of wind directions of interest. The directions are specified in terms of adjacent half-octants, each centered about one of the 16 main compass directions, identified in Table 1, along with their respective code numbers.

Table 1. Compass Directions and Computer Code Numbers

Direction	Code	Direction	Code
NNE	1	SSW	9
NE	2	SW	10
ENE	3	WSW	11
E	4	W	12
ESE	5	WNW	13
SE	6	NW	14
SSE	7	NNW	15
S	8	N	16

For example, if the directions of interest sweep the 22.5° angle centered about the NNE direction, the input to the program is specified as 1,1. If the directions of interest sweep the four adjacent 22.5° angles centered about the S, SSW, SW, and WSW directions, the input to the program is specified as 8,11. If the directions of interest sweep the two adjacent half-octants centered about the N and NNE directions, the input to the program is specified as 16,1. Note that the order of the directions input to the program corresponds to a clockwise sense. A sample interactive session using the program appears in the Appendix.

The output of the program consists of the fastest 60-sec wind speed over open terrain (Exposure Category C as described in ANSI A58-1982 [6]), at the coastline, at 10m above ground, corresponding to the specified directions and to the 10-yr, 25-yr, 50-yr, 100-yr, 500-yr, 1000-yr, and 2000-yr mean recurrence intervals. In addition, the output includes the parameters of the Weibull distribution that best fits the hurricane wind speed data. Knowledge of these parameters allows estimates to be made of speeds corresponding to any mean recurrence interval. The user may also select an optional graphical output consisting of the probability plot of the simulated hurricane wind speed data. Examples of probability plot output are included in the Appendix.

Once the estimated 60-sec wind speeds at 10m elevation, over open terrain,  $v_{open}$ , are known, it is possible to estimate the corresponding 60-sec wind speeds at 10m elevation over water,  $v_{water}$ , as follows [7]:

$$v_{water} \approx 1.26 v_{open} \quad (10)$$

60-sec wind speeds at elevations different from 10m can be estimated as in Ref. 7. To obtain wind speeds averaged over time intervals different from 60 seconds, see Ref. 1, p. 62.

The results obtained by this program could in some instances exhibit minor differences with respect to the results of Ref. 5. Such differences are due to sampling errors inherent in the Monte Carlo simulations on which the respective results are based. The quality of the estimates obtained by the computer program presented here is the same as for the estimates of Ref. 5. Owing to the relatively small climatological data samples upon which the hurricane wind speed estimates at locations north of Cape Hatteras are based, these estimates should be viewed with caution. For comments on the errors in the estimation of hurricane wind speeds, the reader is referred to Ref. 5.

#### 4. ACKNOWLEDGMENTS

The writers wish to thank R.D. Marshall and Bruce Ellingwood, Center for Building Technology, National Bureau of Standards, for their helpful suggestions. We would especially like to thank Dr. James A. Lechner of the Center for Applied Mathematics, National Bureau of Standards, for his enlightening discussions and for his valuable and patient help.

## 5. REFERENCES

1. Simiu, E., and Scanlan, R.H., Wind Effects on Structures, Wiley-Interscience, New York, 1978.
2. Russell, L.R., "Probability Distributions of Hurricane Effects," Journal of the Waterways, Harbors, and Coastal Engineering Division, ASCE, Vol. 97, No. WW1, February, 1971.
3. Tryggvason, B.V., Surry, D., and Davenport, A.G., "Predicting Wind Induced Response in Hurricane Zones," Journal of the Structural Division, ASCE, Vol. 102, No. ST12, December, 1976.
4. Gomes, L., and Vickery, B.J., On the Prediction of Tropical Cyclone Gust Speeds Along the Northern Australian Coast, Research Report R278, School of Civil Engineering, University of Sydney, February, 1976.
5. Batts, M.E., et al, Hurricane Wind Speeds in the United States, NBS Building Science Series 124, National Bureau of Standards, Washington, D.C., May, 1980.
6. American National Standard Minimum Design Loads for Buildings and Other Structures, ANSI A58.1, American National Standards Institute, New York, 1982.
7. Biétry, J., Sacré, C., and Simiu, E., "Mean Wind Profiles and Change of Terrain Roughness," Journal of the Structural Division, ASCE, Vol. 104, No. ST10, October, 1978.






Figure 1. Locator Map

## 6. APPENDIX - PROGRAM LISTING AND EXAMPLES

This appendix includes (1) a listing of the FORTRAN program described in this report, (2) an example interactive computer run using the program, and (3) example graphical output obtained from the program. The examples presented were run to solve the following problem:

Problem: Find (1) the parameters of the Weibull distributions of largest annual hurricane wind speeds and (2) the wind speeds corresponding to various mean recurrence intervals for milepost 2400 (Delaware Bay). Consider winds blowing from (1) all directions, (2) directions ENE, E, ESE, SE, SSE, and S, (3) just direction NE, and (4) all directions from W to E going clockwise.

The graphical output consists of the probability plots for each problem described and was obtained using a Tektronix graphics terminal.

PROGRAM LISTING



PROGRAM WINDS (TAPE5,OUTPUT=/1000)

C THE PURPOSE OF THIS PROGRAM IS TO ESTIMATE COASTLINE HURRICANE WIND SPEEDS  
C CORRESPONDING TO USER SPECIFIED DIRECTIONS AND MEAN RECURRENCE INTERVALS.

C THE PROGRAM CONSISTS OF THIS MAIN ROUTINE AND TWO SUBROUTINES CALLED  
C 'WEIB' AND 'SORT'. INPUT DATA FOR THE PROGRAM ARE ON 56 DATA FILES CON-  
C TAINING SIMULATED DIRECTIONAL HURRICANE WIND SPEED DATA AS DESCRIBED IN  
C REF. 1. THERE IS ONE DATA FILE FOR EACH OF 56 STATIONS OR MILEPOSTS  
C DEFINED ALONG THE GULF AND ATLANTIC COASTS. THE FIRST STATION IS MILE-  
C POST 150 AT PORT ISABEL, TEXAS AND THE LAST STATION IS MILEPOST 2900  
C JUST NORTHEAST OF PORTLAND, MAINE. INTERMEDIATE STATIONS ARE LOCATED  
C AT 50 NAUTICAL MILE INTERVALS BETWEEN MILEPOSTS 150 AND 2900. A LOCATOR  
C MAP SHOWING THE EASTERN SEABOARD AND THE GULF COAST AND THEIR MILEPOSTS  
C IS IN FIG. 1 OF THE REPORT (ALSO, IN REF. 2).

C THIS PROGRAM HAS BEEN WRITTEN ON MAGNETIC TAPE, ALONG WITH 56 DATA FILES  
C CONTAINING SIMULATED HURRICANE WIND SPEED DATA. THE FIRST FILE ON THE  
C TAPE IS THIS FORTRAN PROGRAM. THE 2ND THROUGH 57TH FILES ARE THE DATA  
C FILES CORRESPONDING TO MILEPOSTS 150 TO 2900 (STEP 50), IN ORDER. FOR  
C EXAMPLE, TO ESTIMATE HURRICANE WIND SPEEDS AT MILEPOST 500, COPY THE FIRST  
C FILE (THE PROGRAM) AND THE  $(500-150)/50 + 2 = 9$ TH FILE FROM THE TAPE.  
C THEN RUN THE PROGRAM USING THE SELECTED DATA FILE.

C IF THE STATION OF INTEREST IS NOT LOCATED AT AN INTEGER MULTIPLE OF 50  
C NAUTICAL MILES, THE DATA FILE CORRESPONDING TO THE NEAREST MILEPOST  
C SHOULD BE SELECTED. FOR EXAMPLE, IF IT IS DESIRED TO ESTIMATE HURRICANE  
C WIND SPEEDS AT CAPE HATTERAS - MILEPOST 2170 - THEN THE DATA FILE COR-  
C RESPONDING TO MILEPOST 2150 SHOULD BE SELECTED.

C IT IS SUGGESTED THAT HURRICANE WIND SPEEDS BE ESTIMATED FOR A PARTICULAR  
C MILEPOST BY RUNNING THE PROGRAM FOR THE MILEPOST OF INTEREST, AND FOR THE  
C TWO MILEPOSTS TO EITHER SIDE OF THE SELECTED STATION. THE MEAN RECUR-  
C RENCE INTERVAL WIND SPEEDS CAN THEN BE CALCULATED AS THE AVERAGE OF THE  
C THREE VALUES OBTAINED.

C THE PROGRAM PROMPTS THE USER FOR INPUT TO SPECIFY THE DIRECTIONS OF INTEREST.  
C FOR THIS PROGRAM, THE COMPASS IS DIVIDED INTO 16 HALF-OCTANTS, EACH WITH  
C A 22.5 DEGREE ANGLE CENTERED ABOUT ONE OF THE 16 PRINCIPAL COMPASS  
C DIRECTIONS LISTED BELOW (WITH THEIR RESPECTIVE CODES).

C	1-NNE	5-ESE	9-SSW	13-WNW
C	2- NE	6- SE	10- SW	14- NW
C	3-ENE	7-SSE	11-WSW	15-NNW
C	4- E	8- S	12- W	16- N

C FOR EXAMPLE, IF THE DIRECTION OF INTEREST SWEEPS THE 22.5 DEGREE ANGLE  
C CENTERED ABOUT THE NNE DIRECTION, THE INPUT TO THE PROGRAM IS '1,1'.  
C IF THE DIRECTIONS OF INTEREST SWEEP THE FOUR ADJACENT 22.5 DEGREE ANGLES  
C CENTERED ABOUT THE S, SSW, SW, AND WSW COMPASS POINTS, THE INPUT TO  
C THE PROGRAM IS '8,11'. IF THE DIRECTIONS OF INTEREST ARE FROM N TO NNE,  
C THE INPUT IS '16,1'. NOTE THAT THE ORDER OF DIRECTION INPUT CORRESPONDS  
C TO A CLOCKWISE SENSE OF ROTATION.

C THE OUTPUT OF THE PROGRAM CONSISTS OF THE FASTEST 60-SECOND WIND SPEED  
C OVER OPEN TERRAIN AT THE COASTLINE AT 10 METERS ABOVE GROUND, CORRES-  
C PONDING TO THE SPECIFIED DIRECTIONS AND TO 10-, 25-, 50-, 100-, 500-,  
C 1000-, AND 2000-YEAR MEAN RECURRENCE INTERVALS. IN ADDITION, THE OUTPUT  
C INCLUDES THE PARAMETERS OF THE WEIBULL DISTRIBUTION THAT BEST FIT THE  
C HURRICANE WIND SPEED DATA, ACCORDING TO THE MAXIMUM PROBABILITY PLOT  
C CORRELATION COEFFICIENT CRITERION OF BEST FIT.

C IT IS POSSIBLE TO OBTAIN THE 60-SECOND WIND SPEEDS BLOWING OVER OPEN  
C WATER BY MULTIPLYING THE DATA BY A SUITABLE FACTOR. IT IS SUGGESTED  
C THAT A FACTOR OF APPROXIMATELY 1.26 BE USED (REF. 3).

C TO OBTAIN WIND SPEEDS AT ELEVATIONS OTHER THAN 10 METERS, AGAIN A SUIT-  
C ABLE FACTOR CAN BE USED TO MODIFY THE DATA (SEE REF. 3).

C WIND SPEEDS AVERAGED OVER TIME INTERVALS DIFFERENT FROM 60-SECONDS (FOR  
C EXAMPLE, 1-HOUR) CAN BE CALCULATED BY INCLUDING A FACTOR AS DISCUSSED  
C IN REF. 4, P. 62.

C NOTE THAT THE RESULTS OBTAINED BY THIS PROGRAM WITH THE ASSOCIATED DATA  
C FILES COULD DIFFER INSIGNIFICANTLY FROM THOSE IN REF. 1 BECAUSE OF THE  
C SAMPLING ERRORS THAT ARISE WHEN DIFFERENT MONTE CARLO SIMULATIONS ARE  
C PERFORMED.

C THIS PROGRAM HAS BEEN WRITTEN WITH THE INTENT THAT IT BE EASY TO  
C MODIFY TO SUIT THE PARTICULAR USER'S NEEDS.

C REFERENCES:

C 1. BATTS, M.E., ET AL, "HURRICANE WIND SPEEDS IN THE UNITED STATES,"  
C BUILDING SCIENCE SERIES BSS-124, NATIONAL BUREAU OF STANDARDS,  
C WASHINGTON, DC, 1980.

C 2. BATTS, M.E., RUSSELL, L.R., SIMIU, E., "HURRICANE WIND SPEEDS IN  
C THE UNITED STATES," JOURNAL OF THE STRUCTURAL DIVISION, ASCE,  
C VOL. 106, NO. ST10, OCTOBER, 1980.

C 3. BIETRY, J., SACRE, C., SIMIU, E., "MEAN WIND PROFILES AND CHANGE  
C OF TERRAIN ROUGHNESS," JOURNAL OF THE STRUCTURAL DIVISION, ASCE,  
C VOL. 104, NO. ST10, OCTOBER, 1978.

C 4. SIMIU, E., SCANLAN, R.H., WIND EFFECTS ON STRUCTURES, JOHN WILEY & SONS,  
C NEW YORK, 1978.

```
DIMENSION V(1000,16),VMAX(1000),X(1000),Y(1000)
DIMENSION F(7),MRI(7),VMRI(7),XX(2),YY(2)

REAL LOC

LOGICAL PLTFLAG

DATA MRI /10,25,50,100,500,1000,2000/

PLTFLAG = .FALSE.

READ (5,100) MPOST,N,URATE
100 FORMAT (2I10,F10.0)

NSAVE = N
URATES = URATE

DO 110 I = 1,N
110 READ (5,120) (V(I,J), J=1,16),VMAX(I)
120 FORMAT (16F7.3,F8.5)

130 N = NSAVE

WRITE (*,140)
140 FORMAT (//,' ENTER THE INTEGER CODE FOR THE BEGINNING AND',/,
1      ' ENDING DIRECTIONS (GOING CLOCKWISE)',/,
2      ' 0,0-STOP',/,
3      ' 1-NNE    5-ESE    9-SSW    13-WNW',/,
4      ' 2- NE     6- SE    10- SW    14- NW',/,
5      ' 3-ENE    7-SSE    11-WSW    15-NNW',/,
6      ' 4- E     8- S     12- W     16- N')

READ (*,*) JBGJ,JEND

IF (JBGJ .LE. 0) THEN
WRITE (*,150)
150 FORMAT (/, ' EXIT FROM PROGRAM')
IF (PLTFLAG) CALL DONEPL
STOP
END IF
```

C THE BASIC WIND SPEED ON THE TAPES IS IN KNOTS, AT 10 METERS ABOVE  
C GROUND, AT THE SHORELINE (0 MILES FROM SHORE), OVER OPEN TERRAIN, AND  
C AVERAGED OVER 1 MINUTE.

C WE WANT SPEED IN MPH, AT 10 METERS HEIGHT, AT THE SHORELINE,

C OVER OPEN TERRAIN, AND AVERAGED OVER 1 MINUTE.

C CUNITS = CONVERSION FACTOR FOR CONVERTING BASIC SPEED TO SPEED IN  
C UNITS OF MILES PER HOUR  
C CHIGHT = CONVERSION FACTOR FOR CONVERTING BASIC SPEED TO SPEED AT  
C SPECIFIED HEIGHT  
C CSHORE = CONVERSION FACTOR FOR CONVERTING BASIC SPEED TO SPEED AT  
C SPECIFIED DISTANCE INLAND FROM THE SHORE  
C CROUGH = CONVERSION FACTOR FOR CONVERTING BASIC SPEED TO SPEED OVER  
C SPECIFIED TERRAIN WITH A SPECIFIED ROUGHNESS  
C CTIME = CONVERSION FACTOR FOR CONVERTING BASIC SPEED TO SPEED AVERAGED  
C OVER A SPECIFIED PERIOD OF TIME

CUNITS = 1.152  
CHIGHT = 1.0  
CSHORE = 1.0  
CROUGH = 1.0  
CTIME = 1.0

C = CUNITS\*CHIGHT\*CSHORE\*CROUGH\*CTIME

IF (JBGJ .EQ. 1 .AND. JEND .EQ. 16) THEN

DO 160 I = 1,N  
160 X(I) = C\*VMAX(I)  
  
ELSE IF (JBGJ .EQ. JEND) THEN  
  
DO 170 I = 1,N  
170 X(I) = C\*V(I,JBGJ)  
  
ELSE IF (JBGJ .GT. JEND) THEN  
  
DO 200 I = 1,N  
X(I) = 0.0  
DO 180 J = JBGJ,16  
180 X(I) = AMAX1(C\*V(I,J),X(I))  
DO 190 J = 1,JEND  
190 X(I) = AMAX1(C\*V(I,J),X(I))  
200 CONTINUE

ELSE

DO 220 I = 1,N  
X(I) = 0.0  
DO 210 J = JBGJ,JEND  
210 X(I) = AMAX1(C\*V(I,J),X(I))  
220 CONTINUE

END IF

CALL SORT (X,N,X)

VCUT = 0.0

C WRITE (\*,230)  
C 230 FORMAT (/, ' ENTER THE CUT-OFF VALUE OF WIND SPEED')  
C READ (\*,\*) VCUT

DO 240 I = 1,N  
IF (X(I) .GT. VCUT) GO TO 250  
240 CONTINUE

250 NCUT = I - 1  
N = N - NCUT

C WRITE (\*,260) NCUT  
C 260 FORMAT (' THE NUMBER OF EXCLUDED WIND SPEEDS = ',15)  
C WRITE (\*,270)  
C 270 FORMAT (/, ' ENTER BEGIN AND END INDICES OF X(I) TO BE PRINTED')  
C READ (\*,\*) IBGN,IEND  
C WRITE (\*,280) (X(I), I = IBGN,IEND)

C 280 FORMAT (7F10.2)

URATE = (REAL(N)/REAL(NSAVE))\*URATES

DO 290 I = 1,N

X(I) = X(I+NCUT)

Y(I) = EXP(-URATE\*(N+1-I)/(N+1))

290 CONTINUE

CALL WEIB (N,X,Y,SCALE,SHAPE,LOC)

WRITE (\*,300) MPOST,JBGN,JEND,SHAPE,SCALE,LOC

300 FORMAT (/,

1 ' RESULTS OF FITTING A THREE PARAMETER WEIBULL DISTRIBUTION',/,

2 ' TO EXTREME DIRECTIONAL WIND SPEED DATA',//,

3 ' MILEPOST', I5,/,

4 ' FROM DIRECTION NUMBER',I3,/,

5 ' TO DIRECTION NUMBER ',I3,//,

6 ' SHAPE PARAMETER =',G15.5,/,

7 ' SCALE PARAMETER =',G15.5,/,

8 ' LOCATION PARAMETER =',G15.5)

RSHAPE = 1.0/SHAPE

DO 310 I = 1,7

F(I) = 1.0 - 1.0/MRI(I)

VMRI(I) = LOC + SCALE\*(-ALOG(1.0 - F(I)))\*RSHAPE

310 CONTINUE

WRITE (\*,320) (VMRI(I), I=1,7)

320 FORMAT (//,

1 ' MEAN RECCURENCE INTERVAL WIND SPEEDS',/,

2 ' 10 25 50 100 500 1000 2000',/,

3 7F8.1)

C THE FOLLOWING SUBROUTINE CALLS WILL PLOT A CUMULATIVE FREQUENCY  
C DIAGRAM FOR THE DATA ON WEIBULL PROBABILITY PAPER, AND WILL PLOT  
C THE BEST FIT WEIBULL DISTRIBUTION AS DETERMINED IN SUBROUTINE WEIB.  
C THE SUBROUTINES ARE PART OF THE DISSPLA PLOTTING SOFTWARE LIBRARY.

WRITE (\*,330)

330 FORMAT (' SELECT ONE',/, ' 0-NO PLOT',/, ' 1-PLOT')

READ (\*,\*) IA

IF (IA .EQ. 1) THEN

IF (.NOT. PLTFLAG) CALL TEKALL (4014,960,0,1,0)

PLTFLAG = .TRUE.

CALL PAGE (15,0,12,0)

CALL HEIGHT (0.25)

CALL AREA2D (13,0,8,0)

CALL XNAME ('WIND SPEED (MPH)\$',100)

CALL HEIGHT (0.35)

CALL CARTOG

CALL MX1ALF ('STANDARD', '\*')

CALL MX2ALF ('INSTRU', '%')

CALL YNAME ((' %M5\*LN %M0P1\*N%G1E1M0 \*-%EXM0\* )%M7H0.5E\*1/G\$',100)

CALL HEIGHT (0.18)

CALL MESSAG ('N%G1E1M0\*-%EX\* = MEAN RECURRENCE INTERVAL IN YEARS\$'  
\* ,100,2.0,9.0)

CALL MESSAG ('%M7\*G%M0\* = WEIBULL DISTRIBUTION TAIL LENGTH PARAMET  
\*ER\$',100,2.0,8.5)

XMAX = 0.0

YMAX = 0.0

YMIN = 100.0

DO 340 I = 1,N

Y(I) = (-ALOG(1.0 - Y(I)))\*RSHAPE

YMAX = AMAX1(Y(I),YMAX)

YMIN = AMIN1(Y(I),YMIN)

XMAX = AMAX1(X(I),XMAX)



340 CONTINUE

```
XMAX = 25.0*REAL(INT(XMAX/25.0 + 0.99))
YSTEP = (YMAX - YMIN)/5.0
CALL GRAF (0.0,25.0,XMAX,YMIN,YSTEP,YMAX)
```

```
CALL CURVE (X,Y,N,0)
```

```
XX(1) = 0.0
YY(1) = -LOC/SCALE
```

```
IF (YY(1) .LT. YMIN) THEN
YY(1) = YMIN
XX(1) = SCALE*YMIN + LOC
END IF
```

```
XX(2) = XMAX
YY(2) = (XMAX-LOC)/SCALE
```

```
IF (YY(2) .GT. YMAX) THEN
YY(2) = YMAX
XX(2) = SCALE*YMAX + LOC
END IF
```

```
CALL CURVE (XX,YY,2,0)
```

```
CALL ENDPL (0)
```

```
END IF
```

```
GO TO 130
```

```
END
SUBROUTINE WEIB (N,X,Z,SC,SH,LP)
```

```
REAL LP
```

```
DIMENSION X(1000),Y(1000),Z(1000)
DIMENSION GAMTAB(42),CC(42),SLOPE(42),YINT(42)
```

```
DATA GAMTAB
```

1 /	1.0,	2.0,	3.0,	4.0,	5.0,	6.0,	7.0,	8.0,
2	9.0,	10.0,	11.0,	12.0,	13.0,	14.0,	15.0,	16.0,
3	17.0,	18.0,	19.0,	20.0,	21.0,	22.0,	23.0,	24.0,
4	25.0,	30.0,	35.0,	40.0,	45.0,	50.0,	60.0,	70.0,
5	80.0,	90.0,	100.0,	150.0,	200.0,	250.0,	350.0,	500.0,
6	750.0,1000.0/							

```
DO 140 I = 1,42
```

```
A = 1.0/GAMTAB(I)
```

```
DO 110 J = 1,N
```

```
110 Y(J) = (-ALOG(1.0 - Z(J)))**A
```

```
CALL PPCCAN (N,X,Y,SLOPE(I),YINT(I),CC(I))
```

```
140 CONTINUE
```

C FIND THE DISTRIBUTION WITH THE MAXIMUM PPCC.

```
ISAVE = 1
CCMAX = CC(1)
```

```
DO 400 I = 2,42
IF (CC(I) .GT. CCMAX) THEN
ISAVE = I
CCMAX = CC(I)
END IF
```

```
400 CONTINUE
```

```
SH = GAMTAB(ISAVE)
```

```

SC = SLOPE(ISAVE)
LP = YINT(ISAVE)

RETURN
END
SUBROUTINE PPCCAN (N,X,Y,SLOPE,YINT,CC)

DIMENSION X(1000),Y(1000)

RN = FLOAT(N)

SUM1=0.0
SUM2=0.0

DO 120 J = 1,N
SUM1 = SUM1+X(J)
SUM2 = SUM2+Y(J)
120 CONTINUE

XBAR = SUM1/RN
YBAR = SUM2/RN

SUM1 = 0.0
SUM2 = 0.0
SUM3 = 0.0

DO 130 J = 1,N
SUM1 = SUM1 + (X(J)-XBAR)**2
SUM2 = SUM2 + (Y(J)-YBAR)**2
SUM3 = SUM3 + (X(J)-XBAR)*(Y(J)-YBAR)
130 CONTINUE

SLOPE = SUM3/SUM2
YINT = XBAR-SLOPE*YBAR
CC = SUM3/SQRT(SUM1*SUM2)

RETURN
END
SUBROUTINE SORT (X,N,Y)

```

C THIS SUBROUTINE SORTS A VECTOR, X, IN ASCENDING ORDER AND PUTS THE  
C RESULT IN VECTOR Y.

C FORMAL PARAMETERS:

C RECEIVED:  
C X = VECTOR OF NUMBERS TO BE SORTED.  
C N = NUMBER OF VALUES IN X() AND Y(Y) (DIMENSION OF X AND Y).

C RETURNED:  
C Y = VECTOR OF SORTED NUMBERS.

DIMENSION X(N),Y(N),IU(36),IL(36)

C CHECK THE INPUT VECTOR FOR OBVIOUS SORTS.

```

IF (N .LE. 0) THEN
WRITE (*,100) N
100 FORMAT (' ERROR IN SUBROUTINE SORT, N =', I10)
STOP
END IF

```

```

IF (N .EQ. 1) THEN
Y(1) = X(1)
RETURN
END IF

```

```

X1 = X(1)
Y(1) = X1

```

```

DO 110 I = 2,N
IF (X(I) .NE. X1) GO TO 120

```

```
Y(I) = X1
110 CONTINUE

120 DO 130 I = 2,N
    IF (X(I) .LT. X(I-1)) GO TO 250
    Y(I) = X(I)
130 CONTINUE
```

C START POINT FOR SORT.

```
250 DO 260 I = 1,N
    Y(I) = X(I)
260 CONTINUE

M=1
I=1
J=N
305 IF(I .GE. J)GOTO370
310 K=I
    MID=(I+J)/2
    AMED=Y(MID)
    IF(Y(I) .LE. AMED)GOTO320
    Y(MID)=Y(I)
    Y(I)=AMED
    AMED=Y(MID)
320 L=J
    IF(Y(J) .GE. AMED)GOTO340
    Y(MID)=Y(J)
    Y(J)=AMED
    AMED=Y(MID)
    IF(Y(I) .LE. AMED)GOTO340
    Y(MID)=Y(I)
    Y(I)=AMED
    AMED=Y(MID)
    GOTO340
330 Y(L)=Y(K)
    Y(K)=TT
340 L=L-1
    IF(Y(L) .GT. AMED)GOTO340
    TT=Y(L)
350 K=K+1
    IF(Y(K) .LT. AMED)GOTO350
    IF(K .LE. L)GOTO330
    LMI=L-I
    JMK=J-K
    IF(LMI .LE. JMK)GOTO360
    IL(M)=I
    IU(M)=L
    I=K
    M=M+1
    GOTO380
360 IL(M)=K
    IU(M)=J
    J=L
    M=M+1
    GOTO380
370 M=M-1
    IF(M .EQ. 0)RETURN
    I=IL(M)
    J=IU(M)
380 JMI=J-I
    IF(JMI .GE. 11)GOTO310
    IF(I .EQ. 1)GOTO305
    I=I-1
390 I=I+1
    IF(I .EQ. J)GOTO370
    AMED=Y(I+1)
    IF(Y(I) .LE. AMED)GOTO390
    K=I
395 Y(K+1)=Y(K)
    K=K-1
    IF(AMED .LT. Y(K))GOTO395
```

```
Y(K+1)=AMED  
GOTO390  
END
```

EXAMPLE INTERACTIVE COMPUTER SESSION



@use 7,mp2400.  
READY  
>@xqt wind.f

ENTER THE INTEGER CODE FOR THE BEGINNING AND  
ENDING DIRECTIONS (GOING CLOCKWISE)

0,0-STOP

1-NNE 5-ESE 9-SSW 13-WNW

2- NE 6- SE 10- SW 14- NW

3-ENE 7-SSE 11-WSW 15-NNW

4- E 8- S 12- W 16- N

>1,16

RESULTS OF FITTING A THREE PARAMETER WEIBULL DISTRIBUTION  
TO EXTREME DIRECTIONAL WIND SPEED DATA

MILEPOST 2400

FROM DIRECTION NUMBER 1

TO DIRECTION NUMBER 16

SHAPE PARAMETER = 3.0000

SCALE PARAMETER = 95.106

LOCATION PARAMETER = -72.624

MEAN RECCURENCE INTERVAL WIND SPEEDS

10	25	50	100	500	1000	2000
53.0	67.8	77.2	85.6	102.2	108.5	114.4

SELECT ONE

0-NO PLOT

1-PLOT

>0

ENTER THE INTEGER CODE FOR THE BEGINNING AND  
ENDING DIRECTIONS (GOING CLOCKWISE)

0,0-STOP

1-NNE 5-ESE 9-SSW 13-WNW

2- NE 6- SE 10- SW 14- NW

3-ENE 7-SSE 11-WSW 15-NNW

4- E 8- S 12- W 16- N

>3,8

RESULTS OF FITTING A THREE PARAMETER WEIBULL DISTRIBUTION  
TO EXTREME DIRECTIONAL WIND SPEED DATA

MILEPOST 2400

FROM DIRECTION NUMBER 3

TO DIRECTION NUMBER 8

SHAPE PARAMETER = 1000.0

SCALE PARAMETER = 69248.

LOCATION PARAMETER = -69271.

MEAN RECCURENCE INTERVAL WIND SPEEDS

10	25	50	100	500	1000	2000
35.2	58.4	71.9	83.3	104.0	111.4	118.0

SELECT ONE

0-NO PLOT

1-PLOT

>0

ENTER THE INTEGER CODE FOR THE BEGINNING AND  
ENDING DIRECTIONS (GOING CLOCKWISE)

0,0-STOP

1-NNE 5-ESE 9-SSW 13-WNW  
2- NE 6- SE 10- SW 14- NW  
3-ENE 7-SSE 11-WSW 15-NNW  
4- E 8- S 12- W 16- N

>2,2

RESULTS OF FITTING A THREE PARAMETER WEIBULL DISTRIBUTION  
TO EXTREME DIRECTIONAL WIND SPEED DATA

MILEPOST 2400

FROM DIRECTION NUMBER 2

TO DIRECTION NUMBER 2

SHAPE PARAMETER = 2.0000

SCALE PARAMETER = 53.772

LOCATION PARAMETER = -59.744

MEAN RECCURENCE INTERVAL WIND SPEEDS

10	25	50	100	500	1000	2000
21.9	36.7	46.6	55.6	74.3	81.6	88.5

SELECT ONE

0-NO PLOT

1-PLOT

>0

ENTER THE INTEGER CODE FOR THE BEGINNING AND  
ENDING DIRECTIONS (GOING CLOCKWISE)

0,0-STOP

1-NNE 5-ESE 9-SSW 13-WNW  
2- NE 6- SE 10- SW 14- NW  
3-ENE 7-SSE 11-WSW 15-NNW  
4- E 8- S 12- W 16- N

>12,4

RESULTS OF FITTING A THREE PARAMETER WEIBULL DISTRIBUTION  
TO EXTREME DIRECTIONAL WIND SPEED DATA

MILEPOST 2400

FROM DIRECTION NUMBER 12

TO DIRECTION NUMBER 4

SHAPE PARAMETER = 3.0000

SCALE PARAMETER = 90.160

LOCATION PARAMETER = -67.934

MEAN RECCURENCE INTERVAL WIND SPEEDS

10	25	50	100	500	1000	2000
51.1	65.2	74.1	82.1	97.8	103.8	109.3

SELECT ONE

0-NO PLOT

1-PLOT

>0

ENTER THE INTEGER CODE FOR THE BEGINNING AND  
ENDING DIRECTIONS (GOING CLOCKWISE)

0,0-STOP

1-NNE 5-ESE 9-SSW 13-WNW  
2- NE 6- SE 10- SW 14- NW  
3-ENE 7-SSE 11-WSW 15-NNW  
4- E 8- S 12- W 16- N

>0,0

EXIT FROM PROGRAM

>



EXAMPLE GRAPHICAL OUTPUT






Figure A1. Probability Plot, Directions 1-16




Figure A2. Probability Plot, Directions 3-8




Figure A3. Probability Plot, Direction 2




Figure A4. Probability Plot, Directions 12-4

U.S. DEPT. OF COMM. <b>BIBLIOGRAPHIC DATA SHEET</b> (See instructions)	1. PUBLICATION OR REPORT NO. NBSIR 86-3317	2. Performing Organ. Report No.	3. Publication Date FEBRUARY 1986
---	---	---------------------------------	--------------------------------------

4. TITLE AND SUBTITLE  
  
Directional Hurricane Wind Speeds

5. AUTHOR(S)  
  
Erik M. Hendrickson and Emil Simiu

6. PERFORMING ORGANIZATION (If joint or other than NBS, see instructions)  NATIONAL BUREAU OF STANDARDS DEPARTMENT OF COMMERCE WASHINGTON, D.C. 20234	7. Contract/Grant No.
	8. Type of Report & Period Covered  Research, Final

9. SPONSORING ORGANIZATION NAME AND COMPLETE ADDRESS (Street, City, State, ZIP)  
  
Nuclear Regulatory Commission  
Washington, DC 20555

10. SUPPLEMENTARY NOTES  
  
 Document describes a computer program; SF-185, FIPS Software Summary, is attached.

11. ABSTRACT (A 200-word or less factual summary of most significant information. If document includes a significant bibliography or literature survey, mention it here)

This report presents a simple procedure for estimating coastline hurricane wind speeds corresponding to any specified set of wind directions and to any specified mean recurrence interval. The procedure uses simulated directional hurricane wind speed data as described in the report Hurricane Wind Speeds in the United States (NBS BSS 124). These data are encoded on magnetic tape for 56 mileposts located along the U.S. Gulf and Atlantic coasts. The input to the procedure consists of the location along the coastline and the set of wind directions of interest. The output consists of the fastest 60-sec wind speeds at 10m above ground, in open terrain, corresponding to the set of directions specified in the input and to various mean recurrence intervals. In addition, the output includes the parameters of the Weibull distribution that best fits the hurricane wind speeds.

12. KEY WORDS (Six to twelve entries; alphabetical order; capitalize only proper names; and separate key words by semicolons)

Climatology; hurricanes; statistical analysis; structural engineering; wind engineering

13. AVAILABILITY  <input checked="" type="checkbox"/> Unlimited <input type="checkbox"/> For Official Distribution. Do Not Release to NTIS <input type="checkbox"/> Order From Superintendent of Documents, U.S. Government Printing Office, Washington, D.C. 20402.  <input checked="" type="checkbox"/> Order From National Technical Information Service (NTIS), Springfield, VA. 22161	14. NO. OF PRINTED PAGES  32
	15. Price  \$9.95







